TD3 - Ekouadif

Nicolas Estibals*

8, 15 et 22 janvier 2009

Exercice 1 Quelques équations différentielles

1. Nous cherchons à étudier l'équation différentielle suivante :(E) : $y' + 2xy = e^{x-x^2}$.
   (a) Résolvez (E).
   (b) Résolvez (E) en imposant une condition initiale.
   (c) Tracez sur un même graphe, des solutions de (E) avec différentes conditions initiales.
2. Étudiez de la même manière l'équation (E') : $y'' - 3y' + 2y = xe^x + sin(x)$.
3. Maintenant nous nous intéressons à (E'') : $y'' + 2\sin(y) = 0$.
   (a) Essayez de faire le même type de résolution.
   (b) Devant l'échec, renseignez vous sur la commande `odeplot` et tentez une résolution numérique.
   (c) Créez un programme dépendant du paramètre a qui renvoie le graphe de la solution à (E'') qui passe par 0 et de pente a à l'origine.
   (d) Observez alors sur un même graphe différents "comportements" de cette équation.
   (e) Reconnaissez vous l'équation d'un problème classique ?

Exercice 2 Méthode d'Euler

Comme vous l'avez vu, nous ne sommes pas toujours capable de résoudre les équations différentielles exactement, il faut alors avoir recours au méthodes de résolutions numériques approchées. Nous allons voir dans cette exercice comment approcher les solutions d'une équation différentielle ordinaire, c'est à dire de la forme (E) : $y' = f(x, y(x))$.

1. Soit y la solution de (E) tel que $y(x_0) = y_0$, donnez une approximation au premier ordre de $y(x_0 + h)$. Nous avons ainsi obtenu une approximation d'un nouveau point de la solution y : $(x_1, y_1)i$. Ainsi pour obtenir le graphe d'une solution y de (E) qui passe par (x_0, y_0) sur l'intervalle $[x_0, x_N]$, on peut chercher l'abscisse de $N + 1$ points de la courbe régulièrement écartés (de $h = \ldots$).
2. Écrivez la procédure de prototype `eul(f,x0, xN, N, y0)` qui renvoie la liste de ces $N + 1$ points (sous la forme : `[[x0,y0],[x1,y1], ...,[xN.yN]]`).
3. Tracez sur un même graphe votre approximation et la solution exacte de (E) : $y' = 2x^2y^2$ entre 0 et 1 avec $y(0) = 1$.

Exercice 3 Discontinuité. . .

Trouvez l'unique solution continue sur $\mathbb{R}$ de $(E)xy' + (1-x)y = \frac{xe^x}{x^4+1}$. (Tracer quelques courbes intégrable aidera !)

*N'hésitez pas à m'écrire à l'adresse `Nicolas.Estibals@ens-lyon.fr`!
Vous pouvez retrouver les sujets sur `http://perso.ens-lyon.fr/nicolas.estibals/`.

TD 3 : Ekouadif

Nicolas Estibals (Nicolas.Estibals@ens-lyon.fr, http://perso.ens-lyon.fr/nicolas.estibals)

Exercice 1 : Equations diférentielles

```
> restart;
```

Question 1/ (a)
On peut sauver l'équation dans une variable pour plus de praticité et la résoudre grâce à 'dsolve'.

```
> E:= diff(y(x),x) +2*x*y(x) =exp(x-x**2) ;
  dsolve(E,y(x));
```

$$E := \left(\frac{d}{dx}\,y(x)\right) + 2\,x\,y(x) = \mathbf{e}^{(x-x^2)}$$

$$y(x) = (\mathbf{e}^{x} + \_C1)\,\mathbf{e}^{(-x^2)}$$

Question 1/ (b)
Pour résoudre l'équation E en imposant une condition initiale il faut créer le système d'équation contenant E et la condition initiale.

```
> dsolve({E,y(0)=a},y(x));
```

$$y(x) = (\mathbf{e}^{x} + a - 1)\,\mathbf{e}^{(-x^2)}$$

Question 1/ (c)
Pour tracer un encemble de solution dépendant de la condition initiale, sauvons d'abord l'expression de la solution dans une variable.

```
> f:=rhs(%);
  with(plots): plot( {seq(f,a in {-5,-2,-1,0,1,2,5}) }, x=-2..2);
```

$$f := (\mathbf{e}^{x} + a - 1)\,\mathbf{e}^{(-x^2)}$$

```
Warning, the name changecoords has been redefined
```



Question 2/
La méthode est la même que précédement, attention cependant l'équation est d'ordre 2 il y a donc 2 conditions initiales à fixer.

```
> E1:=diff(y(x),x$2)-3* diff(y(x),x)+2*y(x)=x*exp(x)+sin(x);
  dsolve({E1,y(0)=a,D(y)(0)=b},y(x));
```

$$E1 := \left(\frac{d^2}{dx^2}\,y(x)\right) - 3\left(\frac{d}{dx}\,y(x)\right) + 2\,y(x) = x\,\mathbf{e}^{x} + \sin(x)$$

$$y(x) = -\frac{1}{2}\,\mathbf{e}^{x}\,x^2 - x\,\mathbf{e}^{x} + \frac{3}{10}\cos(x) + \frac{1}{10}\sin(x) + \mathbf{e}^{(2\,x)}\left(-a + \frac{6}{5} + b\right) + \mathbf{e}^{x}\left(-b - \frac{3}{2} + 2\,a\right)$$

```
> g:=rhs(%);
  plot( {seq(seq(g,a in {-10,-2,0,4}) ,b in {0,1})}, x=-1..1,y=-11..11);
```

$$g := -\frac{1}{2}\,\mathbf{e}^{x}\,x^2 - x\,\mathbf{e}^{x} + \frac{3}{10}\cos(x) + \frac{1}{10}\sin(x) + \mathbf{e}^{(2\,x)}\left(-a + \frac{6}{5} + b\right) + \mathbf{e}^{x}\left(-b - \frac{3}{2} + 2\,a\right)$$



Question 3/ (a)

```
> E2:= diff(y(x),x,x) + 2 * sin(y(x))=0;
  eqs2:={E2,y(0)=a,D(y)(0)=b};
  dsolve(eqs2,y(x));
```

$$E2 := \left(\frac{d^2}{dx^2}\,y(x)\right) + 2\sin(y(x)) = 0$$

$$eqs2 := \left\{ D(y)(0) = b, \left(\frac{d^2}{dx^2} y(x) \right) + 2\sin(y(x)) = 0, y(0) = a \right\}$$

$$y(x) = \mathrm{RootOf}\left(\int_0^{\_Z} \frac{1}{\sqrt{4\cos(\_f) - 4\cos(a) + b^2}}\, d\_f - x - \int_0^{a} \frac{1}{\sqrt{4\cos(\_f) - 4\cos(a) + b^2}}\, d\_f \right), y(x) = \mathrm{RootOf}\left(\int_0^{\_Z} \frac{1}{\sqrt{4\cos(\_f) - 4\cos(a) + b^2}}\, d\_f + x - \int_0^{a} \frac{1}{\sqrt{4\cos(\_f) - 4\cos(a) + b^2}}\, d\_f \right)$$

Maple ne peut pas exprimer formellement et complètement les solutions de l'équation (il reste des RootOf et des intégrales qu'il ne sait pas intégrer formellement.)

Question 3/ (b)

Pour les résolutions numériques, on ne peut plus utiliser des variables pour définir les condiions initiales, il faut donner des valeurs numériques! Par défaut 'numpoints=50', cela limite beaucoup la precision du tracé il faut augmenter cette valeur.

```
> s:=dsolve(subs({a=0,b=1},eqs2),y(x),numeric);
  odeplot(s,numpoints=200);
```

$s := \mathbf{proc}(x\_rkf45)\ \ldots\ \mathbf{end\ proc};$



Question 3/ (c)

```
> sol:= proc(v)
    local s,graph:
    s:=dsolve(subs({a=0,b=v},eqs2),y(x),numeric):
    graph:=odeplot(s,[x,y(x)],x=-20..20,numpoints=1000):
    return graph:
  end:
```

Question 4/ (d)

```
> display(seq(sol(v),v in {0,1,2,2.82,2.83,3,5}),view=[-20..20,-10..10])
```



Question 4/ (e)
Il s'agit de l'équation du pendule simple. Les deux comportements que vous pouvez observez correspondent au cas des petites oscillations (régime quasi-sinuosïdale) et du tournoiement (le pendule tourne à vitesse quasi-constante autour de son axe.)

Exercice 2 : Méthode d'Euler

```
> restart;
```

Question 1/
Il s'agit d'écrire le développement limité de y en x0 à l'ordre 1.

```
> y(x0+h)=y(x0)+h*D(y)(x0)+o(h);
  y(x0+h)=y0+h*f(x0,y0)+o(h);
```

$$y(x0+h) = y(x0) + h\,\mathrm{D}(y)(x0) + o(h)$$

$$y(x0+h) = y0 + h\,F(x0, y0) + o(h)$$

Question 2/

```
> eul := proc(f,x0,xN,N,y0)
  local h, points, x, y, i:
  h:=(xN-x0)/N:
  x:=x0; y:=y0:
  points:=[[x0,y0]]:
  for i from 1 to N do
    y:=evalf(y+h*f(x,y)):
    x:=x+h:
    points:=[op(points),[x,y]]:
  end do:
  return points:
  end proc:
```

```
>
```

Question 3/

```
> f:=(x,y)->2*x**2*y**2:
  x0:=0: y0:=1: xN:=1:
  sol_exact:=rhs(dsolve({diff(y(x),x)=f(x,y(x)),y(x0)=y0},y(x)));
```

$$sol\_exact := -\frac{3}{2x^3 - 3}$$

```
> plot({sol_exact,eul(f,x0,xN,5,y0),eul(f,x0,xN,10,y0),eul(f,x0,xN,40,y0
  ,eul(f,x0,xN,100,y0)},x=x0..xN);
```